
Firmware Master-Station (Schüler)

'*******************************************************************************
'MUSEE DES MINES RUMELANGE
'PROJET T3 EC+EE 2010/2011
'BETREUER: JEAN-CLAUDE FELTES
'*******************************************************************************
'MASTER-STATION (RTC, LCD, Terminal, Tristate R/W)
'*******************************************************************************

'uC-INIT------------------------------------------------------------------------

$regfile = "m16def.dat"
$crystal = 16000000
$baud = 9600
$hwstack = 100
$swstack = 100
$framesize = 100

'DISPLAY------------------------------------------------------------------------

Config Lcd = 16 * 4
Config Lcdpin = Pin , Db4 = Portc.3 , Db5 = Portc.2 , Db6 = Portd.7 , Db7 = Portd.6 , E = 
Portc.4 , Rs = Portc.5
Cursor Noblink
Cursor Off

'I2C----------------------------------------------------------------------------

Config Sda = Portc.1
Config Scl = Portc.0

'RTC INIT-----------------------------------------------------------------------

Dim S$ As String * 8
Dim Weekday As Byte
Config Clock = User
Config Date = Dmy , Separator = .

'VARIABLEN DECLARIEREN----------------------------------------------------------

Dim In(7) As String * 10
Dim I As Byte
Dim J As Byte

Dim Slaveseconds As Long
Dim Temperature_outside As String * 10
Dim Temperature_inside As String * 10
Dim Temperature_inside_near As String * 10
Dim Onflag As Byte
Dim Unknow As Byte
Dim Co2value As Word

Dim Commandflag As Bit
Dim Onesecflag As Bit
Dim Masterseconds As Long
Masterseconds = 0
Dim Ok As Byte

'SOFTWARE-UART------------------------------------------------------------------

Open "comd.4:9600,8,n,1" For Output As #1
Open "comd.3:9600,8,n,1" For Input As #2

'TRISTATE-----------------------------------------------------------------------

Config Portd.2 = Output
Portd.2 = 1                             '1 = read

'Timer 1s-----------------------------------------------------------------------

Config Pina.7 = Output                  'LED 1s
Config Timer1 = Timer , Prescale = 1024 , Clear Timer = 1 , Compare A = Disconnect
Ocr1ah = High(15625)
Ocr1al = Low(15625)
Tccr1a = 0
On Compare1a Tim1_1s
Enable Compare1a

Enable Interrupts

'HAUPTPROGRAMM------------------------------------------------------------------

Cls
Commandflag = 1

Print "Date" ;
Print "  " ;
Print "Time" ;
Print "  " ;
Print "Master_Sec" ;
Print "  " ;
Print "Slave_Sec" ;
Print "  " ;
Print "Temp_Out" ;
Print "  " ;
Print "Temp_In";
Print "  " ;
Print "Temp_In_near";
Print "  " ;
Print "Heizung" ;
Print "  " ;
Print "Unknow";
Print "  " ;
Print "Co2"

Do

   If Commandflag = 1 Then
      Gosub Empfangen_slave
   End If

   If Onesecflag = 1 Then               'Zeit anzeigen wenn 1s vorbei
      Gosub Lcdclock
      Onesecflag = 0
   End If

   Ok = Inkey()

   If Ok <> "" Then Gosub Tasten

   Waitms 10

Loop

End

'UNTERPROGRAMME-----------------------------------------------------------------
Tim1_1s:

   Onesecflag = 1                       'Flag für Clock setzen
   Incr Masterseconds

   Toggle Porta.7

   Incr J
   If J = 20 Then                       'jede 15 Sekunden Commandflag setzen
      Commandflag = 1
      J = 0
   End If

Return

'-------------------------------------------------------------------------------
Empfangen_slave:

   'write
   Portd.2 = 0
   Waitms 5

   Print #1 , "*"
   Waitms 10

   'Read
   Portd.2 = 1

   For I = 1 To 7
      Input #2 , In(i)
   Next I

   For I = 1 To 7
      If Left(in(i) , 1) = Chr(10) Then In(i) = Mid(in(i) , 2)

   Next I

   Slaveseconds = Val(in(1))
   Temperature_outside = In(2)
   Temperature_inside = In(3)
   Temperature_inside_near = In(4)
   Onflag = Val(in(5))
   Unknow = Val(in(6))
   Co2value = Val(in(7))
   Gosub Lcd_display
   Gosub Terminal

   Commandflag = 0

Return

'-------------------------------------------------------------------------------
Tasten:

Select Case Ok

       Case "s"
         Input "hh:mm:ss" , S$          'mit Variable arbeiten, da sonst Time$ / Date$ nicht 
gesetzt wird
            Time$ = S$
         Input "dd.mm.yy" , S$
            Date$ = S$

       Case "v"
         Commandflag = 1

End Select
Return

'-------------------------------------------------------------------------------
Lcd_display:

   Locate 3 , 1 : Lcd "                "
   Locate 3 , 1 : Lcd "Temp_Out: " ; Temperature_outside ; "C"
   Locate 4 , 1 : Lcd "                "
   Locate 4 , 1 : Lcd "Co2: " ; Co2value ; "ppm"

Return

'-------------------------------------------------------------------------------

Lcdclock:

   Locate 1 , 1 : Lcd "Datum: " ; Date$
   Locate 2 , 1 : Lcd "Zeit:  " ; Time$

Return

'-------------------------------------------------------------------------------

Terminal:

Print Date$ ;
Print "  " ;
Print Time$ ;
Print "  " ;
Print Masterseconds ;
Print "  " ;
Print Slaveseconds ;
Print "  " ;
Print Temperature_outside ;
Print "  " ;
Print Temperature_inside ;
Print "  " ;
Print Temperature_inside_near ;
Print "  " ;
Print Onflag ;
Print "  " ;
Print Unknow;
Print "  " ;
Print Co2value

Return

'-------------------------------------------------------------------------------
Getdatetime:

   I2cstart

   I2cwbyte &HD0
   I2cwbyte 0
   I2cstop

   I2cstart
   I2cwbyte &HD1
   I2crbyte _sec , Ack
   I2crbyte _min , Ack
   I2crbyte _hour , Ack
   I2crbyte Weekday , Ack
   I2crbyte _day , Ack
   I2crbyte _month , Ack
   I2crbyte _year , Nack
   I2cstop

   _sec = Makedec(_sec)
   _min = Makedec(_min)
   _hour = Makedec(_hour)
   _day = Makedec(_day)
   _month = Makedec(_month)
   _year = Makedec(_year)

Return

'-------------------------------------------------------------------------------
Setdate:

   _day = Makebcd(_day)
   _month = Makebcd(_month)
   _year = Makebcd(_year)

   I2cstart
   I2cwbyte &HD0
   I2cwbyte 4
   I2cwbyte _day
   I2cwbyte _month
   I2cwbyte _year
   I2cstop

Return

'-------------------------------------------------------------------------------
Settime:

   _sec = Makebcd(_sec)
   _min = Makebcd(_min)
   _hour = Makebcd(_hour)

   I2cstart
   I2cwbyte &HD0
   I2cwbyte 0
   I2cwbyte _sec
   I2cwbyte _min
   I2cwbyte _hour
   I2cstop

Return

'-------------------------------------------------------------------------------

